Bipendulum

In this section we demonstrate methods for the study of structural properties of linear systems of ordinary differential equations with rational coefficients, i.e. systems defined over the Weyl algebra of differential operators with respect to time t with rational functions in t as coefficients. We consider the example of a mechanical system called bipendulum which consists of two pendula of length $l1$ respectively $l2$, fixed at the two ends of a bar [Pom01]. We load the package `homalg` and the ring-specific package `Janet` providing procedures for the algebraic analysis of linear systems of partial differential equations.

```
> restart;
> with(Janet):
> with(homalg):
```

First we define the lists of independent and dependent variables for the linear system:

```
> ivar:=[t]; dvar:=[x1, x2, x1t, x2t, u];
```

$$ivar := [t]$$

$$dvar := [x1, x2, x1t, x2t, u]$$

`Janet1` indicates that the package `Janet` will be used with one independent variable. `homalg` will then use the Jacobson normal form for ordinary differential operators with rational coefficients to generate the best basis for a module. This demonstrates how the flexibility of `homalg` can be exploited by using different ring-specific features.

```
> RPJ:=`Janet/homalg`;
```

$$RPJ := Janet/homalg$$

```
> RPJ1:=`Janet1/homalg`;
```

$$RPJ1 := Janet1/homalg$$

The system of the bipendulum is described by equating the following system of ordinary differential expressions to 0:

```
> R:=[diff(x1(t),t)-x1t(t), diff(x2(t),t)-x2t(t),
g/l1*x1(t)+diff(x1t(t),t)+g/l1*u(t),
g/l2*x2(t)+diff(x2t(t),t)+g/l2*u(t)];
```

$$R := [(\tfrac{d}{dt}\,\mathrm{x1}(t)) - \mathrm{x1t}(t),\ (\tfrac{d}{dt}\,\mathrm{x2}(t)) - \mathrm{x2t}(t),\ \frac{g\,\mathrm{x1}(t)}{l1} + (\tfrac{d}{dt}\,\mathrm{x1t}(t)) + \frac{g\,\mathrm{u}(t)}{l1},$$
$$\frac{g\,\mathrm{x2}(t)}{l2} + (\tfrac{d}{dt}\,\mathrm{x2t}(t)) + \frac{g\,\mathrm{u}(t)}{l2}]$$

Here g is the gravitational constant, $x1(t)$ and $x2(t)$ are the positions of the end points of the two pendula at time t and $u(t)$ is the position of the bar at time t. These differential expressions are obtained from a second order ordinary differential system by substituting $x1t$ for the derivative of $x1$ with respect to time t and similarly for the derivative of $x2$ with respect to t. Hence, we consider a first order linear system. The corresponding differential operator, which is expected as input for the `homalg` procedures, can be written as follows:

```
> A:=Diff2Op(R, ivar, dvar);
```

$$A := \begin{bmatrix} [[1,\ [t]]] & 0 & [[-1,\ []]] & 0 & 0 \\ 0 & [[1,\ [t]]] & 0 & [[-1,\ []]] & 0 \\ [[\frac{g}{l1},\ []]] & 0 & [[1,\ [t]]] & 0 & [[\frac{g}{l1},\ []]] \\ 0 & [[\frac{g}{l2},\ []]] & 0 & [[1,\ [t]]] & [[\frac{g}{l2},\ []]] \end{bmatrix}$$

Here each entry is to be interpreted as a linear combination of the differential operators $\frac{d^i}{dt^i}$ represented by $[t, ..., t]$, $i \in \mathbb{Z}_{\geq 0}$. For example, $[[C1, [t, t, t]], [C2, [t], [C3, []]]$ represents the differential operator $C1\frac{d^3}{dt^3} + C2\frac{d}{dt} + C3$. We find a presentation of the module associated with the linear system over the Weyl algebra with rational coefficients, i.e. of the cokernel of A:

```
> M:=Cokernel(A, ivar, RPJ);
```

$$M := [[[[[1,\ []]],\ 0,\ 0] = [0,\ 0,\ [[1,\ []]],\ 0,\ 0],\ [0,\ [[1,\ []]],\ 0] = [0,\ 0,\ 0,\ [[1,\ []]],\ 0],\ [0,\ 0,\ [[1,\ []]]] = [0,\ 0,\ 0,\ 0,\ [[1,\ []]]]],\ [[0,\ [[1,\ [t,\ t]],\ [\frac{g}{l2},\ []]],\ [[\frac{g}{l2},\ [t]]]],\ [[[1,\ [t,\ t]],\ [\frac{g}{l1},\ []]],\ 0,\ [[\frac{g}{l1},\ [t]]]]],\ \text{“Presentation”},\ 3+3\,s+\frac{s^2}{1-s},\ [1]]$$

This presentation uses the above notation for differential operators. A more readable representation of M can be obtained by using the procedure `Pres2Diff` from the package `Janet`:

```
> Pres2Diff(M, ivar, dvar);
```

$$\left[[\_T1(t) = x1t(t),\ \_T2(t) = x2t(t),\ \_T3(t) = u(t)],\ \left[\frac{(\frac{d^2}{dt^2}\,\_T2(t))\,l2 + g\,\_T2(t)}{l2} + \frac{g\,(\frac{d}{dt}\,\_T3(t))}{l2},\ \frac{(\frac{d^2}{dt^2}\,\_T1(t))\,l1 + g\,\_T1(t)}{l1} + \frac{g\,(\frac{d}{dt}\,\_T3(t))}{l1}\right],\ \text{“Presentation”},\ 3+3\,s+\frac{s^2}{1-s},\ [1]\right]$$

Using 'Janet1' (and hence the Jacobson normal form) it turns out that the cokernel is cyclic and even free (of rank 1). This only holds in the generic case $l1 \neq l2$ because in the following computation $l1 - l2 \neq 0$ is assumed:

```
> Pres2Diff(Cokernel(A, ivar, RPJ1), ivar, dvar);
```

$$[[\_T1(t) = -\frac{l1\,x1(t)}{l2} + x2(t)],\ [0],\ \text{“Presentation”},\ \frac{1}{1-s},\ [1]]$$

Now we study the case that the lengths $l1$, $l2$ of the pendula are equal:

```
> R2:=subs(l2=l1, R);
```

$$R2 := [(\frac{d}{dt}\,x1(t)) - x1t(t),\ (\frac{d}{dt}\,x2(t)) - x2t(t),\ \frac{g\,x1(t)}{l1} + (\frac{d}{dt}\,x1t(t)) + \frac{g\,u(t)}{l1},\ \frac{g\,x2(t)}{l1} + (\frac{d}{dt}\,x2t(t)) + \frac{g\,u(t)}{l1}]$$

The system needs to be converted to the differential operator form for the use of `homalg`:

```
> A2:=Diff2Op(R2, ivar, dvar);
```

$$A2 := \begin{bmatrix} [[1,\ [t]]] & 0 & [[-1,\ []]] & 0 & 0 \\ 0 & [[1,\ [t]]] & 0 & [[-1,\ []]] & 0 \\ [[\frac{g}{l1},\ []]] & 0 & [[1,\ [t]]] & 0 & [[\frac{g}{l1},\ []]] \\ 0 & [[\frac{g}{l1},\ []]] & 0 & [[1,\ [t]]] & [[\frac{g}{l1},\ []]] \end{bmatrix}$$

Again we find a presentation of the module associated with the linear system over the Weyl algebra with rational coefficients, i.e. of the cokernel of $A2$:

```
> M2:=Cokernel(A2, ivar, RPJ): Pres2Diff(M2, ivar, dvar);
```

$$\left[[\_T1(t) = x1t(t),\ \_T2(t) = x2t(t),\ \_T3(t) = u(t)],\ \left[\frac{(\frac{d^2}{dt^2}\,\_T2(t))\,l1 + g\,\_T2(t)}{l1} + \frac{g\,(\frac{d}{dt}\,\_T3(t))}{l1},\ \frac{(\frac{d^2}{dt^2}\,\_T1(t))\,l1 + g\,\_T1(t)}{l1} + \frac{g\,(\frac{d}{dt}\,\_T3(t))}{l1}\right],\ \text{“Presentation”},\ 3+3\,s+\frac{s^2}{1-s},\ [1]\right]$$

From this presentation the structural properties of the module are not evident at first sight. However, the Jacobson normal form provides a different presentation with two generators, a torsion and a free one:

```
> Pres2Diff(Cokernel(A2, ivar, RPJ1), ivar, dvar);
```

$$\left[[\_\mathrm{T1}(t) = -\mathrm{x1}(t) + \mathrm{x2}(t),\ \_\mathrm{T2}(t) = \mathrm{x1}(t)],\ \left[\frac{(\frac{d^2}{dt^2}\,\_\mathrm{T1}(t))\, \mathit{l1} + g\,\_\mathrm{T1}(t)}{\mathit{l1}} \right],\ \text{"Presentation"},\ 2 + 2\,s + \frac{s^2}{1-s},\ [1] \right]$$

In fact, using 'Janet' again, we find that the torsion submodule of the cokernel of $A2$ is generated by the difference of the positions $x1(t)$, $x2(t)$ of the end points of the two pendula, which is an autonomous element of the system. This autonomous element satisfies the second order ordinary differential equation given in the second entry of the result:

```
> Pres2Diff(TorsionSubmodule(M2, ivar, RPJ), ivar, dvar);
```

$$\left[[\_\mathrm{T1}(t) = \mathrm{x1t}(t) - \mathrm{x2t}(t)],\ \left[\frac{(\frac{d^2}{dt^2}\,\_\mathrm{T1}(t))\, \mathit{l1} + g\,\_\mathrm{T1}(t)}{\mathit{l1}} \right],\ \text{"Presentation"},\ 1 + s,\ [0] \right]$$

Author: Daniel Robertz

Date:	`2005-12-31`
Last modified:	`2007-06-22 17:47`

References

[BCG+03] Y. A. Blinkov, C. F. Cid, V. P. Gerdt, W. Plesken, and D. Robertz, *The* MAPLE *Package* Janet*: I. Polynomial Systems. II. Linear Partial Differential Equations*, Proc. 6th Int. Workshop on Computer Algebra in Scientific Computing, Passau, Germany, 2003, (http://wwwb.math.rwth-aachen.de/Janet).

[BR] Mohamed Barakat and Daniel Robertz, `homalg` – *A meta-package for homological algebra*, submitted. `arXiv:math.AC/0701146` and (http://wwwb.math.rwth-aachen.de/homalg).

[BR07] ______, `homalg` *project*, 2004-2007, (http://wwwb.math.rwth-aachen.de/homalg).

[Pom01] J.-F. Pommaret, *Partial differential control theory*, Mathematics and its Applications, vol. 530, Kluwer Academic Publishers Group, Dordrecht, 2001, Vol. I. Mathematical tools; Vol. II. Control systems, With a foreword by Jan C. Willems. MR MR1853969 (2002h:93001) 1

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: daniel@momo.math.rwth-aachen.de